
Sammanfattning

Den här rapporten undersöker automatisk lösning av Heyawakepussel
genom att använda Answer Set Programming. Heyawake är på många
sätt likt det mer välkända Sudoku men har inte blivit lika efterforskat
vilket gör spelet väldigt intressant att undersöka.

Spelet är bevisat NP-fullständigt så fokus har legat på att imple-
mentera lösaren istället för att göra den så snabb som möjligt men en
del optimeringar har ändå gjorts.

Answer Set Programming ha länge använts för att lösa NP-fullständiga
problem och för Heyawake är det möjligt att lösa normalstora pussel
väldigt snabbt. För att optimera lösaren krävs djupgående kunskaper
om Answer Set Programming-lösaren som används.

Abstract

This report exmamines automatic solving of Heyawake puzzles using
Answer Set Programming. Heyawake is somewhat similar to the more
well known Sudoku but has not been as extensively researched which
makes the game very interesting to examine.

The game has previously been proven to be NP-complete so the focus
has been on implementing the solver instead of making it fast although
some optimizations has been made.

Answer Set Programming has long been used to solve NP-complete
problems and for Heyawake it is possible to solve normal sized puzzles
very fast. Optimizing the solver proved to be hard unless you have ex-
tensive knowledge of the Answer Set Programming solver that is used.

Chapter 1

Introduction

In recent years the Japanese game of Sudoku has become very popular. These
little brain teasing puzzles has become an everyday encounter and can nowa-
days be found in most Swedish newspapers and different cell phone apps. For
someone who has solved a lot of these puzzles, they tend to get kind of simple
and loses their challenge. Fortunately, Sudoku is not the only puzzle of its
kind.

1.1 What is Heyawake?

Heyawake (jap: Divided Rooms]) is a Japanese logical puzzle developed by
Nikoli, the same company that created Sudoku. It is similar to Sudoku in some
ways, but totally different in others. Both games are based on a rectangular
grid of cells which are to be numbered. In Sudoku the grid consists of nine
times nine cells, in Heyawake it can be of any size. A Sudoku puzzle is divided
into nine square rooms of three times three cells, Heyawake is divided into any
number of rooms of any rectangular size. In Sudoku the numbering of cells
goes from one to nine and in Heyawake it only goes from zero to one, making it
a binary-determination puzzle. Similarly to many other binary-determination
puzzles like Bag and Light-Up, the numbering is replaced by the colors white
and black instead of the numbers zero and one to make it clearer for the player
what is going on.

The numbering, or coloring, of the puzzle is to be done according to a set
of rules. In Sudoku the numbers cannot conflict with each other according to
two very similar rules:

• Each row and column has to hold the numbers one through nine.

• Each room has to hold the numbers one through nine.

In Heyawake there are five rules to follow, each of which is very different to
the others:

• Black cells may not be placed immediately next to another black cell
orthogonally.

• All white cells must be orthogonally connected to each other throughout
the entire grid.

• The different rooms can have a number indicating how many black cells
the sub grid must contain.

• Rooms with no number can contain any number of black cells, but the
other rules must still be followed.

• A straight line of white cells must not pass through more than two rooms.

The fifth rule is the most defining of the puzzle as it states the main goal
when solving a Heyawake by hand, which is to prevent any lines of white cells
from passing through three different sub grids. The other rules just have to
be followed while doing this. For a seasoned Sudoku solver Heyawake can
often prove to be an interesting challenge as many puzzles require complex
combinations of the rules to solve them.

Usually when playing Heyawake, different difficulties are defined as different
sizes of the puzzle grid. Harder puzzles tend to have larger grids than easier
ones, the actual difficulty however when solving a puzzle by hand comes from
how hard it is to apply the different rules in a given situation. A smaller puzzle
may be much more difficult than a bigger one, it is all determined by how
easily each cells color can be concluded from the current situation. This means
that actual easier puzzles often require two or less rules in any given situation
to determine the color of a cell whilst harder puzzles require a combination of
more rules to determine what cells to color next.

Determining the difficulty of puzzles according to how easy the actually can
be solved is not done easily and it is not clear if a computer implementation
would find the same puzzles hard that a human would when solving by hand.
In this report a puzzle is considered harder if it has a larger grid.

1.2 Solving a Heyawake puzzle by hand

Below follows a small introduction of how to solve a Heyawake puzzle by hand
since many people may not be familiar with how to play the game.

These are the most basic moves when playing the game.

1. Usually, you start by filling in what is given. Rooms with the number
zero are obviously not going to have any colored cells in them so you can
immediately consider those uncolored as in figure 1.1.

2. There are often rooms that can be filled in because it only has one possible
coloring. For example as seen in figure 1.2, a 3x1 room that needs two
colored cells can only have colored cells at the edges since colored cells
can not be neighbours.

3. When continuing to work on the puzzle the two Heyawake rules that are
most often used to determine what cells to color next. Using rule five, a
colored cell can be placed if a straight line of white cells is about to pass
through three rooms like in figure 1.3.

4. Since colored cells can not be next to each other, according to rule one,
all colored cells can immediately be surrounded by non colored cells like
in figure 1.4.

Figure 1.1: Given rooms can be com-
pleted immediately

Figure 1.2: Some rooms only has one
possible coloring

Figure 1.3: A colored cell at the end
of a straight line of non colored cells
according to rule 5

Figure 1.4: A colored cell must be sur-
rounded orthogonally by non colored
cells

Figure 1.5: The marked cells have to be white

5. The final example, figure 1.5 shows a more advanced way to determine
what cells to be colored. The marked cells in the bottom row has to
be non colored since the non colored cells in the bottom left would not
be able to be connected to all other non colored cells. This might not
be immediately obvious to new players but is an important move when
trying to solve harder puzzles.

1.3 NP-completeness of solving a Heyawake

The problem of solving a Heyawake puzzle has been proven to be NP-complete
[3], which means that it is probably not solvable in polynomial time. Because
of the complexity of the problem, a solver will either have to make sacrifices
to the accuracy (i.e. not solve the puzzle 100% of the time) to get faster, or
spend a lot more time on solving the larger puzzles. Sacrificing accuracy in a
solver would make the solver useless so the only solution here is to spend a lot
of time on the larger puzzles. However, since common Heyawake puzzles rarely
get bigger than 30x30 cells it should be possible to write a solver that runs in
a reasonable timeframe for normal case puzzles.

The goal of this project is project is to create a solver for Heyawake puzzles
and measure how long it takes for the solver to complete puzzles of various
difficulty levels.

1.4 Problem formulation

Can a Heyawake solver be implemented that solves common size puzzles within
one minute or less?

What would such an implementation look like, and is it possible to optimize
the algorithm so that it finds the solutions quicker?

Method

In order to answer the problem a solver must first be implemented and then a
series of tests must be run to determine if the solver works as intended. After
that more tests have to be conducted to test the efficiency of the solver. The
biggest obstacles for the implementation were:

• Choosing the proper programming language.

• Creating a text based representation of the puzzle grid that is applicable
with the chosen programming language.

• Implementing the rules of the puzzle and applying them to different puz-
zles.

2.1 Choosing the proper implementation language

Programming the solver required a language with sufficient capabilities to for-
mulate the rules while at the same time it should be practical for implementing
the grid and applying the constraints. An imperative language like C would
require a lot of iteration, as would most kinds of object oriented approaches.
There would need to be checks for all the rules over the whole grid and the
overall complexity of the code would be way too large to be practical.

Using a declarative programming style like constraint programming or an-
swer set programming, that is about specifying what to be computed rather
than how to compute it, to solve the task is more practical when it comes to
these kind of problems [5]. If the grid is implemented in a proper way the
Heyawake rules could be implemented as constraints or rules for the coloring
of the grid which would make the implementation cleaner and simpler than the
imperative approach.

In constraint programming (CP) the goal is to write constraints between
variables that acts as restrictions for them. The constraints should represent
the characteristics of the solution which can then be used by the solver to find
a solution. The constraints often take the form of mathematical comparisons
between variables but can be much more complex than that.

In the case of Heyawake the constraints would be the rules of the game and
a solver would be instructed to fill in all cells of the grid within the constraints.

5

Answer Set Programming (ASP) is a form of declarative programming
where the goal is to first represent some data and then create a set of rules
from which, when applied to the data, an answer set can be deduced. ASP is
primarily oriented towards solving NP-hard search problems [4]. ASP is often
a bit more specific about what the solver has to do to reach the solution but is
still centered around telling the solver what to reach and not how to do it [1].

In the case of Heyawake the ASP rules would be the rules of Heyawake and
the answer set is the unique solution to a certain puzzle.

Both approaches are viable ways of trying to solve the puzzle but since ASP
offer ways to easily express conditional effects and advanced constructs [1],
as well as being oriented towards NP-hard problems, the ASP approach was
chosen in this project.

An ASP solution is often composed by a grounder and a solver [4]. The
grounder removes all variables and outputs ground terms that can be used by
the solver to compute an answer set. The three most commonly used grounders
for ASP programming are DLV, LParse1 and GrinGo, which were written for
the solver DLV, SModels and Clasp respectively [2].

While DLV acts both as a grounder and a solver, LParse and GrinGo can
be used to ground ANSProlog into ground terms that can be used by most
solvers [2]. In a series of tests run by Petritsch, where he compared the com-
putational run times between DLV, LParse with Smodels/Clasp and Gringo
with Smodels/Clasp, he showed that GrinGo in conjunction with Clasp pro-
vided the fastest computations [6]. Therefore we have chosen to implement our
solver with GrinGo and Clasp.

2.2 Creating a text based representation of the
Heyawake grid in ANSProlog

A Heyawake grid consists of NxM cells. The grid is divided into a certain
number of subgrids which we will call rooms. Since these rooms occupy the
entire grid, there is only need for a notation that describes all rooms, and the
grid will follow naturally from that.

A room has three properties, its size, its location and an optional number.
Since all rooms are rectangular, both the size and location can be derived from
a rooms coordinates in the grid (col1, row1, col2, row2) where (col1, row1) rep-
resents the top left corner of the room and (col2, row2) represents the bottom
right corner. A fifth parameter (number) is needed for the rooms optional
numbers2. It could also be useful to be able to refer to a room by an ID so
that will be the final parameter. A final representation of a room written in
ANSProlog can be seen in Listing 2.1.

Listing 2.1: A room described in ANSProlog
room( Id , Col1 , Row1 , Col2 , Row2 , Number ) .

1ANSProlog (ANswer Set PROgramming LOGic) can refer to all ASP languages but can
also be used to specifically refer to the most commonly used ASP language LParse (which
in turn can be used by most grounders). In this report we will use ANSProlog as a name for
the LParse language (not for the grounder) and ASP languages for the more general case of
ASP languages

2The number -1 is used to represent the rooms where no number is given

Using only the rooms as input to the grounder works but by also including
the width and height of the grid, instead of having to calculate the size from the
rooms alone, the grounder can more easily create the cells needed for solving
the puzzle which makes the whole program quicker.

2.3 Implementing the rules of the puzzle

A short introduction to ANSProlog
For the sake of those not familiar with ANSProlog, a quick guide to ANSProlog
is included below.

The syntax of ANSProlog is somewhat similar to more well known con-
straint programming languages like Prolog. ANSProlog is known as a declara-
tive programming [1] lanugage which means that the whole language is oriented
around atoms which represent all data in a program.

When writing a program in ANSProlog you either write atoms, rules that
atoms can be derived from, or rules limiting what atoms can exist. The
grounder takes the code and presents it in a highly logical data structure using
mostly numbers. The solver can then interpret the data and create an answer
set consisting of the atoms that were described by the rules.

When setting up the rules there are a number of built in logical function-
alities that can be used.

• Rules
head :− body .

A rule is very similar to rules in other logical programming languages.
The head consists of some data you want to represent and the body is
the constraint that forces the head to take a certain form. All atoms in
the body must be true for the head to exist.

Listing 2.2: Example of a rule
notEqual (X,Y) :− X! = Y.

Listing 2.2 shows a way of implementing the rule X must not be equal to
Y.
The body can be omitted from a rule. In this case the head gets no
restriction of how to represent the data, so a rule without a body repre-
sents raw data. Since no uncertain variables are allowed in ANSProlog,
this means that the head must consist of atoms without variables or with
fixed variable values.

Listing 2.3: Example of a rule without a body
something .
something ( 1 , 4 ) .

• Integrity constraints
An integrity constraint is a set of rules that can not be true at the same
time. Listing 2.4 show how something1, something2, and something3 can

not all be true at the same time. This can be applied to both rules and
atoms but must still adhere to the rule of no uncertain variables.

Listing 2.4: Example of an integrity constraint
:− something1 , something2 , something3 .

This rule is often used to prohibit different atoms from existing together.

• Choice functions

Listing 2.5: Syntax of choice rules
X{something1 , . . . , somethingn}Y

The choice function forces the solver to choose something, either for an
answer set or for a dependancy. Listing 2.5 shows the syntax of a choice
rule which forces the solver to choose at least X and at most Y variables
of something1 . . . somethingn.
It is important to note that choosing exactly 0 (i.e. X=Y=0) items are
impossible using this rule, and has to be worked around. By ommitting
X and Y from the rule, any number of items can be chosen.

Listing 2.6: Example of a choice rule
1{ something1 , something2 , something3 }2

Listing 2.6 means choose at least one and at most two of the atoms
something1, something2, and something3.

The solver in this report will only be using different applications of the rules
mentioned above.

ANSProlog implementation of the Heyawake rules
Constraints to describe the puzzle

• Create all the cells of the grid as atoms based on the width and height
of the room.

Listing 2.7: Initialize all cells of the grid
c e l l ( 0 . . (X−1 ) , 0 . . (Y−1)):−width (X) , he ight (Y) .

• A cell must be either black or white. This constraint forces the answer
set to contain the answer we seek, which is a representation of the grid
in only black and white cells.

Listing 2.8: Choose exactly one of white or black for every cell
1{white (X,Y) , b lack (X,Y)}1:− c e l l (X,Y) .

• A cell is always in a certain room

Listing 2.9: Make sure that all cells are in a room
inroom (X,Y,R):− c e l l (X,Y) , room(R,X1 ,Y1 ,X2 ,Y2 ,N) ,X<=X2 ,X1<=X,Y<=Y2 ,Y1<=Y.

Describe inroom(X,Y,R) for every cell(X,Y) where the number R is the
ID of the room where the cell belongs.

• There are a fixed number of rows and columns in the grid (0-indexed)

Listing 2.10: Create all the columns and rows of the grid
colum ( 0 . .X−1):−width (X) .
row ( 0 . .Y−1):−he ight (Y) .

• A cell is always next to 2-4 other cells (orthogonally)

Listing 2.11: Make sure that all neighbours are neighbours
nextTo (X1 ,Y,X2 ,Y):−X1=X2+1, c e l l (X1 ,Y) , c e l l (X2 ,Y) .
nextTo (X,Y1 ,X,Y2):−Y1=Y2+1, c e l l (X,Y1) , c e l l (X,Y2 ) .
nextTo (X1 ,Y,X2 ,Y):−X1=X2−1, c e l l (X1 ,Y) , c e l l (X2 ,Y) .
nextTo (X,Y1 ,X,Y2):−Y1=Y2−1, c e l l (X,Y1) , c e l l (X,Y2 ) .

For every cell, another cell is nextTo that cell if it has coordinates next
to it.

The rules of the game

• Rule 1: Colored cells may not be immediately next to another
cell This rule is easily described using an integrity constraint.

Listing 2.12: If two cells are black, they can not be neighbours
:−nextTo (X1 ,Y1 ,X2 ,Y2) , b lack (X1 ,Y1) , b lack (X2 ,Y2 ) .

• Rule 2: All white cells must be connected to each other This rule
requires that the generated nextTo atoms are extended into connected
atoms which determine if a white cell is connected to another white cell.
Connected either means that two white cells are next to each other or
that a white cell is connected to some third white cell which is next to
the second white cell.

Listing 2.13: All white cells must be connected to each other
%Two white c e l l s are connected e i t h e r i f they
%are immediate ly next to each other
connected (X1 ,Y1 ,X2 ,Y2):−nextTo (X1 ,Y1 ,X2 ,Y2) , white (X1 ,Y1)

, white (X2 ,Y2 ) .
%Two white c e l l s are connected i f one o f them i s next to a
%white t h a t i s connected to the o ther .
connected (X1 ,Y1 ,X2 ,Y2):− connected (X1 ,Y1 ,X3 ,Y3) ,

nextTo (X2 ,Y2 ,X3 ,Y3) , white (X1 ,Y1) , white (X2 ,Y2) , white (X3 ,Y3 ) .

Finally, an integrity constraint is added so that no two white cells can
exist if they are not connected.

Listing 2.14: No two white cells can not be connected
:−white (X1 ,Y1) , white (X2 ,Y2) ,not connected (X1 ,Y1 ,X2 ,Y2 ) .

• Rule 3: A room with a number must have equally many black
cells in it as the number indicates

Listing 2.15: For rooms with the number N, choose N black cells
N{black (X,Y) : c e l l (X,Y) :X<=X2 :X1<=X:Y<=Y2 :Y1<=Y}N:−

room(R,X1 ,Y1 ,X2 ,Y2 ,N) ,N>0.

The choice function can not be used when trying to choose zero items of
something. Therefore, in rooms that require zero black cells, choose K
white cells within the room where K is the number of cells in the room.

Listing 2.16: Make rooms with 0 all white
K{white (X,Y) : c e l l (X,Y) :X<=X2 :X1<=X:Y<=Y2 :Y1<=Y}K:−

room(R,X1 ,Y1 ,X2 ,Y2 ,N) ,N=0,K=(X2−X1+1)∗(Y2−Y1+1).

• Rule 4: Blank rooms can contain any number of colored cells
The first implementation included listing 2.17 which were meant to make
sure that a blank room contained colored cells where needed. However,
it was removed as the code was unneccesary after making sure that all
cells need to be black or white. The removal of this listing is later used
as the optimization of our code.

Listing 2.17: Blank rooms contain any number of colored cells
{ black (X,Y) : c e l l (X,Y) :X<=X2 :X1<=X:Y<=Y2 :Y1<=Y}:−

room(R,X1 ,Y1 ,X2 ,Y2 ,N) ,N=−1.

• Rule 5: An orthogonal line of white cells must not pass through
more than two rooms

Listing 2.18: Make sure white lines do not cross to many rooms
%For rows in the g r i d
1{ black (C1 . . A3 ,Y)}:−

3{ inroom (X1 ,Y,R) , inroom (X2 ,Y,R) , inroom (X3 ,Y,R3)}3 ,
c e l l (X1 ;X2 ;X3 ,Y) ,
row (Y) ,
B1<=Y,Y<=D3,
R1!=R2 ,R1!=R3 ,R2!=R3 ,
room(R1 ,A1 ,B1 ,C1 ,D1 , E1 ) , room(R2 ,A2 ,B2 ,C2 ,D2 , E2 ) , room(R3 ,A3 ,B3 ,C3 ,D3 , E3 ) ,
C1<A2 ,C2<A3 .

%For columns in the g r i d
1{ black (X,D1 . . B3)} :−

3{ inroom (X,Y1 ,R1) , inroom (X,Y2 ,R2) , inroom (X,Y3 ,R3)}3 ,
c e l l (X,Y1 ;Y2 ;Y3) ,
column (X) ,
A1<=X, X<=C3 ,
R1!=R2 , R1!=R3 , R2!=R3 ,
room(R1 ,A1 ,B1 ,C1 ,D1 , E1 ) , room(R2 ,A2 ,B2 ,C2 ,D2 , E2 ) , room(R3 ,A3 ,B3 ,C3 ,D3 , E3 ) ,
D1<B2 , D2<B3 .

Choose at least one colored cell so that when you choose exactly three
rooms that create a line of cells on the checked row or column, the colored
cell is placed either in the middle room or on the edges of the ’outer’ rooms
to prevent the line from going through all the rooms.

2.4 Testing the implementation

The output when running Clasp with the ground terms generated by GrinGo is
the answer set that the solver could find as well as two separate runtimes, the
computational runtime and the total runtime. In most cases the computational
runtime is less than 1/10 of the total runtime, which is probably a consequence
of the massive input that clasp has to read before it can start to compute an
answer set.

The computational time is very interesting as this value shows how com-
plex the puzzle has been to solve for the computer while the total runtime is
interesting because if the software is going to be usable the total runtime can
not be too large. Due to this, both computational runtime and total runtime
will be measured and compared for the different puzzles.

Three difficulty levels of puzzles is to be used during the testing, denoted in
our source of puzzles as easy, medium and hard. Easy puzzles are 6x6 cells big,
medium 9x9, and hard 11x11. Ten puzzles from each of the different difficulty
levels have been collected and the solver will be run on the puzzles ten times
each.

As an optimization attempt, the code in listing 2.17 was removed and as
such, the same tests will be run on the code both with and without the code
in 2.17.

The puzzles used are taken from the publicly available android/iOS app
’Heyawake’ by Paul Wagener and are used with permission.

All tests were run on an Asus ZenBook ux31a with a 4-core intel i7 processor
at 2.4 GHz.

Results

The solver solves all of our test cases correctly.
The average computational time and total runtime, across all tests, before

optimization the attempt can be seen in figure 3.6 and after the optimization
attempt in figure 3.7.

A comparison of the results before and after the optimization attempt can
be seen in figure 3.5.

The results show that our optimization attempt seems successful since the
average time for solving the puzzles were lower almost across the board. They
also show that medium sized puzzles took strictly longer time to solve than
easy puzzles and hard puzzles took strictly longer time to solve than medium
puzzle.

Using exponential regression tools in MatLab an exponential curve was
fitted to the values for the two different run times, as seen in figure 3.1 and
figure 3.2. The R-square value represents the likelihood of that the plotted
values actually exist in an exponential relation and the curve indicates that
the run times actually grows exponentially with the grid size.

Figure 3.1: Expected total run time on an NxN grid

13

Figure 3.2: Expected computational time on an NxN grid

Figure 3.3: Total run time Figure 3.4: Computational time

Figure 3.5: Comparison of runtimes before and after optimization. Each dot
represents the average run time for a puzzle. In both cases easy puzzle values
are placed to the left, hard in the middle and medium to the right in the curve.
Blue is before optimization and red is after.

Problem Computational runtime (s) Total runtime (s)
Easy 1 0.03 0.1103
Easy 2 0.03 0.1021
Easy 3 0.03 0.1138
Easy 4 0.03 0.1263
Easy 5 0.02 0.1132
Easy 6 0.02 0.1092
Easy 7 0.03 0.1142
Easy 8 0.03 0.1133
Easy 9 0.03 0.1502
Easy 10 0.02 0.0881
Medium 1 0.20 3.066
Medium 2 0.21 3.3617
Medium 3 0.17 4.7031
Medium 4 0.21 2.7959
Medium 5 0.18 1.4884
Medium 6 0.20 4.1783
Medium 7 0.22 1.9012
Medium 8 0.20 3.7047
Medium 9 0.19 2.9257
Medium 10 0.25 2.9500
Hard 1 0.59 19.3581
Hard 2 0.54 15.4820
Hard 3 0.49 18.9715
Hard 4 0.47 18.1408
Hard 5 0.48 30.0787
Hard 6 0.48 23.9885
Hard 7 0.53 15.0605
Hard 8 0.50 32.2746
Hard 9 0.90 15.3994
Hard 10 0.59 14.6988

Figure 3.6: Average run times of all test puzzles before the optimization at-
tempt

Problem Computational runtime (s) Total runtime (s)
Easy 1 0.02 0.1016
Easy 2 0.03 0.1146
Easy 3 0.03 0.0901
Easy 4 0.02 0.1190
Easy 5 0.02 0.0959
Easy 6 0.02 0.0969
Easy 7 0.02 0.1060
Easy 8 0.02 0.1234
Easy 9 0.04 0.1016
Easy 10 0.03 0.0898
Medium 1 0.19 3.3989
Medium 2 0.19 3.4327
Medium 3 0.17 4.4920
Medium 4 0.22 2.6349
Medium 5 0.16 1.4176
Medium 6 0.28 4.7950
Medium 7 0.19 1.6986
Medium 8 0.20 2.9061
Medium 9 0.21 2.9061
Medium 10 0.31 3.0177
Hard 1 0.55 18.3288
Hard 2 0.52 15.0331
Hard 3 0.48 17.3074
Hard 4 0.46 16.5323
Hard 5 0.46 28.8938
Hard 6 0.45 22.3667
Hard 7 0.53 14.6894
Hard 8 0.50 30.8753
Hard 9 0.88 14.8652
Hard 10 0.58 13.3518

Figure 3.7: Average run times of all test puzzles after the optimization attempt

Discussion

We have implemented a solver for Heyawake puzzles that seems to be able to
solve any heyawake puzzle given sufficient time. We do not have any proof
that the solver works correctly all the time but it solves all of our test puzzles
correctly, and with that fact we will draw the conclusion that the solver is
correct.

4.1 Efficiency of the Solver

The time taken to actually solve a puzzle seems to grow exponentially when the
grid size gets larger. According to the results of our exponential regression the
measured points exist in exponential relation to each other with almost 100%
certainty. This is not surprising at all given that the problem is NP-complete
and no known algorithm can solve a problem of that difficulty in polynomial
time.

Even though the solver was able to complete all its task within our tar-
get timeframe we were not completely satisfied with all of the total runtimes.
Both medium and easy difficulties could be solved in reasonable times. If this
software were to be used by someone, a few seconds of waiting is acceptable.
However if one were to solve hard puzzles and had to wait up to 30 seconds
that could possibly be a reason not to use this software.

4.2 Further Optimizations

By omitting our special code for the blank room check we have shown that
optimizations are possible to make. In almost every test the solver was no-
ticeably faster after removing this unnecessary code. For the hard puzzles the
results were unanimous, the solver was faster without the blank room code. In
the two lower difficulties the results were a bit more unclear, we believe this
to be due to errors in the timing. Assuming that there is some marginal error
in clasp’s timing of the runs, this error will have greater effect in the shorter
solving times than the longer. Because of this it is reasonable to believe that
the measurements for our hard puzzles are more likely to reflect the actual
solving times for our puzzles.

Having made this optimization we get the indication that further optimiza-
tions can still be made to the solver to bring down the total solving time overall.

17



18 CHAPTER 4. DISCUSSION

With enough work the runtimes could possibly be pushed down far so that even
the hard puzzles can be solved in negligible time frames.

One possible optimization is to add a “first step” that generates the correct
black and white atoms for the rooms that are “given”, to our solver. This
could possibly eliminate unnecessary computation since our solver currently
handles the given rooms in the same way as every other room. Another possible
optimization might be to remove two of the nextTo rules since two of them are
implicit given the other two, X-1 and X+1 most likely generate the same result
if X can be chosen arbitrarily, and therefore they might be unnecessary work
for the grounder when creating the ground terms. Our implementation of the
five rules are the constraints that take the most time to compute. Since the
other constraints mainly consist of generation of atoms without conflicts while
they will most likely be run iteratively, but since the rules can conflict with
each other their results will take a lot longer time to compute. This means that
the rules can probably be implemented in different ways, which could result in
a different runtime than our tests.

4.3 Other implementation options

We feel comfortable with our choice of an ASP approach to the solver. It was
easy to formulate the problem and simple to get the desired results. We don’t
believe choosing the the Constraint Programming approach would have offered
any advantages, the logic is similar but having used constraint programming
in prolog earlier we do not see how it could have offered any easier way to
implement the solver. We don’t see any other advantages such as improved
optimization capabilities either.

The one different approach that may have been more interesting to look at is
implementation using an imperative or object oriented language. This approach
would no doubt be a lot more difficult to implement, since it would require far
more code that our chosen ASP approach. The advantage that could be gained
with this approach however is more control over how the solver operates, which
is something we actively chose to sacrifice when choosing a logical programming
language.

Using an imperative language implementing our “first step” optimization
mentioned above could be done easily. We could also utilize the “human”
methods of solving Heyawake more which might be more effective for the easy
puzzles which does not require advanced usage of the Heyawake rules, but
would probably not be as effective for the harder puzzles since it would require
a lot of iteration whereas a logical solver would not. The most interesting
approach would have been to try an imperative approach on the easy puzzles
with an ASP/CP solution for the harder ones. Or one where the imperative
solver starts off solving the first part and the ASP solution takes over when
the easy deductions are made.



Chapter 5

Conclusions

It is possible to create a solver that can solve many ordinary sized puzzles
within a minute. When the solver takes upwards 30 seconds to complete the
moderately sized puzzles (11x11) (hard difficulty in the report), the users might
get impatient so more optimization would be necessary if this software were to
be used.

Although we are satisfied with our choice of an ASP approach to the prob-
lem we would have liked to see better solving times. The actual normal solving
times for hard puzzles (15-30 s) were a bit higher than we had initially hoped.
During the development we did not believe that any of the tests would run for
over 10 seconds due to the relatively small size of the puzzles (121 cells in a
hard puzzle).

We did show that optimizations were possible, but we still believe more op-
timizations could have been done. These optimizations would however require
a lot of knowledge in how gringo and clasp works since we can not control in
what way the solver works this is a big obstacle to overcome because of the
sparse documentation of the tools.

Chapter 6

Appendix

Note that the full appendix can be found at http://www.csc.kth.se/~cwinsnes/heyawakeappendices.
All tested puzzles and timed tests can be found there.

6.1 Solver code

%Describes the rules of a heyawake game.
%To see how rooms are described, refer to room1.lp.
%
%The five rules of heyawake:
%1. Painted cells may never be orthogonally connected

%2. All white cells must be interconnected (only orthogonal connections count).

%3. A number indicates exactly how many painted cells there are to be in that room.

%4. A room with no number can contain any number of painted cells, including none.

%5. Where a straight orthogonal line of white cells are formed, it must not contain
% cells from more than two rooms.

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Set up
%All cells must be within the grid
cell(0..(X-1),0..(Y-1)) :- width(X), height(Y).

%All cells are in a room
inroom(X, Y, R) :- cell(X, Y), room(R, X1, Y1, X2, Y2, N), X<=X2, X1<=X, Y<=Y2, Y1<=Y.

% All cells are either black or white
1 { white(X,Y), black(X,Y) } 1 :- cell(X,Y).

% Create a column variable for each column
column(0..X-1) :- width(X).

% Create a row variable for each row

row(0..Y-1) :- height(Y).

% RULE 1! %
% Painted cells may never be orthogonally interconnected %
:- nextTo(X1,Y1,X2,Y2), black(X1,Y1), black(X2,Y2).
%:- black(X,Y), black(X-1, Y).
%:- black(X,Y), black(X+1, Y).
%:- black(X,Y), black(X, Y-1).
%:- black(X,Y), black(X, Y+1).

% RULE 2 %
% All white cells must be interconnected
nextTo(X1,Y,X2,Y) :- X1=X2+1, cell(X1,Y), cell(X2,Y).
nextTo(X,Y1,X,Y2) :- Y1=Y2+1, cell(X,Y1), cell(X,Y2).
nextTo(X1,Y,X2,Y) :- X1=X2-1, cell(X1,Y), cell(X2,Y).
nextTo(X,Y1,X,Y2) :- Y1=Y2-1, cell(X,Y1), cell(X,Y2).

connected(X1,Y1,X2,Y2) :- nextTo(X1,Y1,X2,Y2), white(X1,Y1), white(X2,Y2).
connected(X1,Y1,X2,Y2) :- connected(X1,Y1,X3,Y3), nextTo(X2,Y2,X3,Y3), white(X1,Y1), white(X2,Y2), white(X3,Y3).
:- white(X1,Y1), white(X2,Y2), not connected(X1,Y1,X2,Y2).

%Rule 3
%Select all (size of room) white cells for a room with the number 0
%Select all cells(X,Y) where (X,Y) is within the room and make sure they are white.
K{white(X,Y) : cell(X,Y) : X<=X2 : X1<=X : Y<=Y2 : Y1<=Y}K :- room(R, X1, Y1, X2, Y2, N), N=0, K=(X2-X1+1)*(Y2-Y1+1).

%Select N black cells for a room with the number N
%Select all cells(X,Y) where (X,Y) is within the room and make sure that N of them are black.
N { black(X,Y) : cell(X,Y) : X<=X2 : X1<=X : Y<=Y2 : Y1<=Y} N :- room(R, X1, Y1, X2, Y2, N), N>0.

%Rule 4
%Select an arbitrary amount of black cells for a room with blank number (-1)
%{ black(X,Y) : cell(X,Y) : X<=X2 : X1<=X : Y<=Y2 : Y1<=Y} :- room(R, X1, Y1, X2, Y2, N), N=-1.

% RULE 5! %
% Straight orthogonal line of white cells must not contain cells from more than two rooms.
% Choose at least one black cell between the endpoints of three rooms in a row
1{black(C1..A3, Y)} :-
3{inroom(X1,Y,R1),inroom(X2,Y,R2),inroom(X3,Y,R3)}3,
cell(X1;X2;X3,Y),
row(Y),
B1<=Y, Y<=D3,
R1!=R2, R1!=R3, R2!=R3,
room(R1,A1,B1,C1,D1,E1),room(R2,A2,B2,C2,D2,E2), room(R3,A3,B3,C3,D3,E3),
C1<A2, C2<A3.

% Choose at least one black cell between the endpoints of three rooms in a column
1{black(X,D1..B3)} :-
3{inroom(X,Y1,R1),inroom(X,Y2,R2),inroom(X,Y3,R3)}3,
cell(X,Y1;Y2;Y3),
column(X),
A1<=X, X<=C3,
R1!=R2, R1!=R3, R2!=R3,
room(R1,A1,B1,C1,D1,E1),room(R2,A2,B2,C2,D2,E2), room(R3,A3,B3,C3,D3,E3),
D1<B2, D2<B3.

6.2 Example Puzzle Code

Easy puzzle nr 1
%Input format: Width, Height, Room1, Room2...
%
%Width is described by: width(X).
%Height is descibed by: height(Y).
%A room is described by: room(X1,Y1,X2,Y2,N)
% Where X1, Y1 is the top left corner of the room
% X2, Y2 is the bottom right corner of the room
% N is the number of black cells in the room
% (-1 means any number of black cells in the room)
%
%The grid is zero-indexed.

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% EXAMPLE GRID 1 %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%easy 1

width(6).
height(6).
room(1,0,0,3,0,2).
room(2,0,1,2,2,1).
room(3,0,3,2,3,2).
room(4,0,4,2,5,2).
room(5,4,0,5,1,0).
room(6,3,1,3,1,-1).
room(7,3,2,4,3,0).
room(8,5,2,5,3,-1).
room(9,3,4,5,5,2).


